Click Game part 2

Step 3: Show Score

Our game is coming along great! Now let's add some text to display the score on the screen, instead of the messages printing to the console.

For this we need:

1. A new variable to keep track of the score.
2. A new method to display the score, I'll call mine `showScore()`. This should display text on the screen.
3. Call the `showScore()` method in our draw function.
4. Change our `mouseRelesed()` method to add and subtract from the score for hits and misses.

Our code now looks like this:

```
int circleX;
int circleY;
int circleSize;
int score = 0;

void setup() {
  size(800, 600);
}

void draw() {
  background(0);
  showScore();
  showCircle();
}

void showCircle() {
  circleX = width/2;
  circleY = height/2;
  circleSize = 100;

  fill(255, 0, 0);
  circle(circleX, circleY, circleSize);
}

void mouseReleased() {
  if (dist(mouseX, mouseY, circleX, circleY) < circleSize/2) {
    score++;
  } else {
    score--;
  }
}

void showScore() {
  fill(255); // make's text white
  textSize(24);
  textAlign(CENTER);
  text("Score: " + score, width/2, 50);
}
```

Step 4: Make the Circle Blink

Now we need to make the circle appear and disappear every second. We can use a Processing variable called `frameCount`, which keeps track of how many frames have passed since our program started running. Since `draw()` updates at a rate of 60 frames a second, that means that every 60 frames is one second.

We can use the modulus operator `%` to test if the **frameCount** is a multiple of 60. This will only happen once every second.

To do this we need:

1. A `boolean` variable to track whether our circle should be shown or invisible.
2. Create a new method to toggle this variable every 60 frames (1 second).
3. Ad an `if` to the `showCircle` method, so that it only display the circle when its set to visible.

Our code will now look like this:

```
int circleX;
int circleY;
int circleSize;
int score = 0;
boolean circleVisible = false;

void setup() {
  size(800, 600);
}

void draw() {
  background(0);
  showScore();
  toggleCircle();
  showCircle();
}

void toggleCircle() {
  if (frameCount % 60 == 0) {
    circleVisible = !circleVisible;
  }
}

void showCircle() {
  if (circleVisible) {
    circleX = width/2;
    circleY = height/2;
    circleSize = 100;

    fill(255, 0, 0);
    circle(circleX, circleY, circleSize);
  }
}

void mouseReleased() {
  if (dist(mouseX, mouseY, circleX, circleY) < circleSize/2) {
    score++;
  } else {
    score--;
  }
}

void showScore() {
  fill(255); // make's text white
  textSize(24);
  textAlign(CENTER);
  text("Score: " + score, width/2, 50);
}
```

Challenges

1. Can you make the circle a random color each time it blinks?
2. Make a target instead of a circle, and give 5 points for a bulls eye, 2 points for the next ring, and 1 point for the outer ring.
3. Make the screen blink red if you miss.